

# Approximating Politics

12 September 2004  
Eric Kemp-Benedict

## Introduction

The outcome of political negotiation is notoriously difficult to predict, yet some researchers have claimed success in doing just that, using models suggested by Social Choice theory. This note describes a partial implementation of a model created by Bueno de Mesquita (as described in Bueno de Mesquita, 2002) and implemented by Feder (1987). Essentially, Bueno de Mesquita's model starts with the insight of the Marquis de Condorcet that when people vote along a single policy axis, there tends to be one outcome that would beat any of the others if it were paired against them one by one. This "Condorcet winner" is also the preferred outcome of the median voter, and is the outcome that is expected to prevail. In Bueno de Mesquita's model, "people" are generalized to "actors," who might be individuals, states, organizations, etc., and the model is further enriched by generalizing the idea of "votes" in political negotiations. In the model, an actor's "vote" is weighted by how much influence the actor has on the outcome – the actor's *capacity* – the importance of the issue to the actor – the issue's *salience* – and the relative satisfaction the actor derives from one option relative to another – the relative *utility*. These features are present in the model presented in this case study, although it differs in some details from Bueno de Mesquita and Feder's models.

Bueno de Mesquita's full model has both a static and a dynamic component. The dynamic component allows actors to strategically change their revealed positions, to devote greater resources to the problem (reflected by an increased salience) and to try to influence each other to change position. In this note only the static component of the model is implemented. Despite this limitation, the model should be of value. Essentially, variables that are endogenous in the full model remain exogenous in the partial model presented here, which allows users of the model to explore different strategies for coping with an unsatisfactory outcome.

An explicit scenario is presented in this case study, in which the model described here is implemented using the IPAT-S scenario scripting language. The results of running the scenario are discussed. The full script is included as an appendix to this case study, and it can also be downloaded from the IPAT-S web site. Readers who are interested in exploring and modifying the scenarios can download the script and the free IPAT-S software from <http://ipats.kb-creative.net/>.

Note that the model described in this case study should be used with caution, despite the considerable success the Buena de Mesquita model has had in practical applications (Bueno de Mesquita, 2002). The success reported by Bueno de Mesquita (2002) and Feder (1987) is for applications of the full model, not for the truncated model presented here, and successful application appears to be dependent on expert input and critical review. Moreover, although these models appear to be an exception, rational choice and related models in general have a poor record when confronted with empirical evidence (Green and Shapiro, 1994). Nevertheless, this model offers an opportunity to display some obscure tricks for the IPAT-S language. It must also be admitted that, for this author at least, the model is fun to play around with, which was enough reason to share it.

## The Model

Suppose there are  $M$  actors faced with  $N$  possible policy outcomes. An actor  $i$  will derive a level of satisfaction  $u_{ij}$  from seeing outcome  $j$  prevail. (The letter  $u$  stands for “utility,” which is the conventional terminology, but in the rest of this case study it will be referred to as “satisfaction.” It is then easy to imagine, for example, presenting actors with a questionnaire in which they are asked to rank their satisfaction with different outcomes, with  $u_{ij} = 0$  representing “very unsatisfied,”  $u_{ij} = 5$  “neutral” and  $u_{ij} = 10$  “very satisfied”.) An actor brings to the table a capacity  $c_i$  to influence the outcome, and a level of desire to influence the outcome reflected in the issue’s salience,  $s_i$ , for actor  $i$ .

Faced with a choice between an option  $j$  and an option  $k$ , an actor will cast a “vote”  $v_{ijk}$  equal to:

$$v_{ijk} = c_i s_i (u_{ij} - u_{ik}) . \quad (1)$$

If  $v_{ijk}$  is positive, then actor  $i$  prefers  $j$  to  $k$ . If negative, then he or she prefers  $k$  to  $j$ , and if equal to zero, then  $i$  is indifferent to the outcome. The net vote  $V_{jk}$  between  $j$  and  $k$ , taking all actors into account, is determined by summing over  $i$ ,

$$V_{jk} = \sum_{i=1}^M v_{ijk} . \quad (2)$$

If  $V_{jk}$  is positive, then option  $j$  wins over  $k$ . Furthermore, if  $j$  prevails over *any*  $k$  in pairwise competition, then  $j$  is the Condorcet winner, and is the predicted outcome of the negotiations, barring any strategic repositioning on the part of the negotiating actors.

Equations (1) and (2) are the essence of the static portion of the Bueno de Mesquita model, and their simplicity is very satisfying. One difference between the presentation here and the Bueno de Mesquita and Feder models is that in their models the policy axis has numbers associated with it, and the utility is estimated by calculating the distance between an actor’s preferred position and a possible outcome. This seems unnecessarily artificial, as long as the number of possible policy outcomes and actors is not excessively large, since it should be possible for a knowledgeable analyst to assign relative satisfaction (or utility) to different outcomes for the various actors.<sup>1</sup>

## A Specific Implementation

In the specific implementation of the model presented for purposes of illustration in this case study, four actors negotiate over three possible policy outcomes. In keeping with the original focus of the IPAT-S language – sustainable development scenarios – the actors are the *Environment Ministry*, the *Rest of Government*, *Advocacy Groups* that agitate on the issue, and *Industry*. The different policy positions are given stereotypical titles for illustrative purposes: *Command and Control*, *Market Incentive* and *Laissez-Faire*. It is assumed that the Advocacy Groups and Industry are at opposite ideological poles, and will only support Command and Control or Laissez-Faire, respectively. The branches of government are more moderate. The Environment Ministry, because of its historical role, would prefer Command and Control, but is neutral with respect to the Market Incentive option. It is completely opposed to the Laissez-Faire approach. The

---

<sup>1</sup>This is not true in the dynamic model, where a prescription for the utility function is needed to endogenize actor’s strategies.

Rest of Government, which is dominated by an ideology quite divergent from the Environment Ministry, would prefer to see a Laissez-Faire approach, but would be quite satisfied ( $u = 7$ ) with a Market Incentive option. It would be quite unhappy with a Command and Control approach.

In this implementation, it is assumed that “Rest of Government” – that is, government excluding the Environment Ministry – has the greatest capacity to influence the outcome. It is assigned a capacity of 1.0. The Environment Ministry has the second-greatest influence, since the issue falls within its brief. However, since it is somewhat out of odor with the Rest of Government, its capacity is only 0.5. Industry has a relatively large influence over the decision, with a capacity of 0.4, due to the support it has provided to the government in the last election, and because it is small, well-funded and well-organized. The Advocacy Groups, although energetic, do not have as much influence because of a lack of coordination and a negative perception of them within the government; they have a capacity of 0.1.

The values for satisfaction and capacity are assumed to stay the same throughout the scenario, which extends from 2004 to 2010. However, salience changes for some actors. The salience for the Advocacy Groups remains high, at 1.0, throughout the scenario – after all, this is the *raison d'être* for the groups. Salience also remains steady for the Environment Ministry, which has a lot on its plate, at 0.6. However, for Industry and for the Rest of Government salience increases over time. In 2004, the issue is barely noticed by the Rest of Government, with a salience of 0.2. It is more important for Industry, which has a salience of 0.4, but still not as high as it is for either the Ministry or the Advocacy Groups. However, by 2010, both Industry and the Rest of Government have placed the issue high on their agendas, with a salience of 0.8 for Industry and 0.6 for the Rest of Government.

The result is that in 2004 the preferred position is *Command and Control*, but already by 2007 it has switched to *Laissez-Faire*, where it remains. Once the Rest of Government and Industry begin to pay attention to the issue, the weight that the Environment Ministry and the Advocacy Groups can bring to the issue are insufficient to maintain their preferred position.

## Implementing the Model in IPAT-S

The model described in this case study is quite different from the type of models that IPAT-S was originally designed for. Nevertheless, it can be implemented in IPAT-S through the use of some tricks that, while somewhat obscure, are generalizable to other cases. The full script is given in the appendix.

The most easily generalizable trick is that whenever an equation involves a matrix with two identical indices, it can be implemented in IPAT-S by creating a duplicate dimension. In this case, the matrix with identical indices is  $V_{jk}$ , for which the indices  $j$  and  $k$  both refer to policy positions. In IPAT-S this can be implemented by creating two dimensions with identical values, most efficiently implemented by using the IPAT-S *ditto* feature:

```
ditto 'Command and Control' 'Market Incentive' 'Laissez-Faire':  
dim position ''  
dim position_dup ''
```

Then the variable corresponding to  $V_{jk}$  is assigned both the *position* and *position\_dup* dimensions:

```
# Sum over actors to get the pairwise balance of votes  
var Vote{position, position_dup}  
:: Capacity * Salience * (Satisfaction - Satisfaction_dup) -> Vote
```

The expression in the third line of this code fragment illustrates the compact IPAT-S notation. In this expression, *Capacity*, *Saliency*, *Satisfaction* and *Satisfaction\_dup* all contain an *actor* dimension, while the *Vote* variable does not. Because IPAT-S automatically sums over any unmatched dimensions, this is an implementation of Equations (1) and (2) combined.

A final trick is required to determine the Condorcet winner. In this case, the goal is to find a *position* for which *Vote* is positive or zero for any *position\_dup*. This is accomplished using a `foreach` loop to examine *Vote* for each *position\_dup* and check whether it is positive or zero using the IPAT-S `step()` function:

```
# Find the Condorcet winner (who wins in all pairwise groups)
# At the end of the loop, winner = 1 for the Condorcet winner
var winner{position}
# Initialize to 1 for all years
winner = 1
foreach position_dup:
  :: step(Vote{position_dup = ?}) * winner -> winner
:foreach
```

## Conclusion

The challenge of modeling political negotiations is daunting, and there are few successes to point to. One approach, based on Social Choice theory, has reportedly had great success (Bueno de Mesquita, 2002; Feder, 1987). This case study described how a version of the static part of the full Bueno de Mesquita model can be implemented in IPAT-S by using some relatively obscure IPAT-S tricks involving duplicate dimensions, automatic summation, looping and the `step()` function.

While the model itself should be used with caution in practical applications, it is amusing to experiment with, because the results are directly translatable into real-world terms, and because the interaction between capacities, saliency and utilities is difficult to anticipate and can be surprising.

## References

Bueno de Mesquita, B. 2002. *Predicting Politics*. Columbus, Ohio: The Ohio State University Press.

Feder, S.A. "FACTIONS and Policon: New Ways to Analyze Politics," in H.B. Westerfield, ed. 1995. *Inside CIA's Private World: Declassified Articles from the Agency's Journal, 1955-1992*. New Haven: Yale University Press. Reprinted from: *Studies in Intelligence* 31(1), pp. 41-57. 1987.

Green, D.P. and I. Shapiro. 1994. *Pathologies of Rational Choice Theory: A Critique of Applications in Political Science*. New Haven: Yale University Press.

## Appendix: IPAT-S Script

```
base year 2004
scenario years 2007 2010

##
## Define dimensions & mapping between dimensions
##
dim actor 'Environment Ministry' 'Rest of Government' 'Advocacy Groups' 'Industry'
ditto 'Command and Control' 'Market Incentive' 'Laissez-Faire':
dim position ''
dim position_dup ''

var pos_ident{position, position_dup}
ditto 'Command and Control':
pos_ident{position='', position_dup='' } = 1
ditto 'Market Incentive':
pos_ident{position='', position_dup='' } = 1
ditto 'Laissez-Faire':
pos_ident{position='', position_dup='' } = 1

##
## Define initial characteristics of actors
##
# Capacity: power to enforce one's preferences
# Salience: How important the issue is to the actor
# Satisfaction: How much satisfaction the actor gains if the position is adopted
# 0: very unsatisfied
# 5: neutral
# 10: very satisfied
var Satisfaction{actor, position} Capacity{actor} Salience{actor}
ditto actor="Environment Ministry":
Capacity.0{' '} = <0.5>
Satisfaction.0{' ' position = "Command and Control"} = <10>
Satisfaction.0{' ' position = "Market Incentive"} = <5>
Satisfaction.0{' ' position = "Laissez-Faire"} = <0>
ditto actor="Rest of Government":
Capacity.0{' '} = <1.0>
Satisfaction.0{' ' position = "Command and Control"} = <0>
Satisfaction.0{' ' position = "Market Incentive"} = <7>
Satisfaction.0{' ' position = "Laissez-Faire"} = <10>
ditto actor="Advocacy Groups":
Capacity.0{' '} = <0.1>
Satisfaction.0{' ' position = "Command and Control"} = <10>
Satisfaction.0{' ' position = "Market Incentive"} = <0>
Satisfaction.0{' ' position = "Laissez-Faire"} = <0>
ditto actor="Industry":
Capacity.0{' '} = <0.4>
Satisfaction.0{' ' position = "Command and Control"} = <0>
Satisfaction.0{' ' position = "Market Incentive"} = <0>
Satisfaction.0{' ' position = "Laissez-Faire"} = <10>

# Let satisfaction & capacity remain at their base-year values
byv Satisfaction
byv Capacity

# Salience changes over time
var bySalience{actor} finSalience{actor}
ditto actor="Environment Ministry":
bySalience.0{' '} = <0.6>
finSalience.0{' '} = <0.6>
ditto actor="Rest of Government":
bySalience.0{' '} = <0.2>
finSalience.0{' '} = <0.6>
ditto actor="Advocacy Groups":
bySalience.0{' '} = <1.0>
finSalience.0{' '} = <1.0>
ditto actor="Industry":
bySalience.0{' '} = <0.4>
finSalience.0{' '} = <0.8>

# Interpolate between the base-year and final values for Salience
byv bySalience
byv finSalience

:: bySalience + (finSalience - bySalience) * (y - y0)/(y.fin - y.0) -> Salience
```

```
# Make a copy of the "Satisfaction" variable
var Satisfaction_dup{actor, position_dup}
:: Satisfaction * pos_ident -> Satisfaction_dup

# Sum over actors to get the pairwise balance of votes
var Vote{position, position_dup}
:: Capacity * Salience * (Satisfaction - Satisfaction_dup) -> Vote

# Find the Condorcet winner (who wins in all pairwise groups)
# At the end of the loop, Winner = 1 for the Condorcet winner
var Winner{position}
# Initialize to 1 for all years
Winner = 1
foreach position_dup:
  :: step(Vote{position_dup = ?}) * Winner -> Winner
:foreach

report Winner as "Condorcet Winner"
```